package org.as3search.criteria {
	/*
	 * as3search - http://code.google.com/p/as3search
	 * Copyright (c) 2009-2010 by respective authors. All rights reserved.
	 *
	 * This library is free software; you can redistribute it and/or modify it under the
	 * terms of the GNU Lesser General Public License as published by the Free Software
	 * Foundation; either version 2.1 of the License, or (at your option) any later
	 * version.
	 *
	 * This library is distributed in the hope that it will be useful, but WITHOUT ANY
	 * WARRANTY; without even the implied warranty of MERCHANTABILITY or FITNESS FOR A
	 * PARTICULAR PURPOSE. See the GNU Lesser General Public License for more details.
	 *
	 * You should have received a copy of the GNU Lesser General Public License along
	 * with this library; if not, write to the Free Software Foundation, Inc.,
	 * 59 Temple Place, Suite 330, Boston, MA 02111-1307 USA
	 */
	
	/**
	 * A <code>Conjunction</code> is an implementor of <code>Junction</code>
	 * which expresses a compound "AND" condition. It is 
	 * essentially the same concept as the "&amp;&amp;" construct native to AS3. It
	 * allows for multiple conditions to be expressed as necessary in a compound
	 * <code>Criterion</code>. 
	 * 
	 * @example The following code blocks demonstrate what <code>Conjunction</code>
	 * accomplishes:
	 * <listing version="3.0">
	 * new Conjunction()
	 * 	.add(Restrictions.eq("id", 3))
	 * 	.add(Restrictions.eq("name","mary"));
	 * //yields the same comparison as
	 * if (object.id == 3 &amp;&amp; object.name == "mary") {
	 * 	
	 * }
	 * </listing>
	 */
	public class Conjunction implements Junction {
		
		private var criteria:Array = [];
		/**
		 * Creates a new <code>Conjunction</code> instance.
		 */
		public function Conjunction() {
			
		}
		/**
		 * @inheritDoc
		 */
		public function add(value:Criterion):Junction {
			criteria.push(value);
			return this;
		}
		/**
		 * @inheritDoc
		 */
		public function evaluate(value:Object):Boolean {
			for each (var criterion:Criterion in criteria) {
				if (!criterion.evaluate(value))
					return false;
			}
			
			return true;
		}
	}
}